

TO:

# COLOUR GENIE

## AUCKLAND DISTRICT USER GROUP

NEWSLETTER NO. 16

MAY, 1985

---

Hi,  
Another good turnout at the last meeting. Andy was back again, and started off the meeting by showing a pulled-apart Genie to everyone, describing what goes where, and which does what. Chris held his last class on Basic Programming. These have been a great success and at times have had more people in them than the ones in the main room.

Andy mentioned during his talk that your Colour Genie should always be kept covered when it is not being used. He said it was surprising just what was inside both the Genies and the recorders that he has been fixing. Cigarette ash is diabolical - don't smoke while you are on your computer.

The Group are now repairing Colour Genies and have come to an agreement with Rakon to do this. There are several advantages in this, we will do it far cheaper than Rakon, and far quicker! As you do when you send your machine to Rakon, you will just have to courier it to Ken or Andy at their address in Whenuapai. They will fix it, then courier it back to you with an invoice for the cost. Auckland members can arrange to leave their machines with me, either at work or at home, and I will ring them and let them know when their machines are fixed. By the way, recorders can be repaired too, again at a very reasonable charge.

### DISK DRIVES

---

The Disk Drive interface and DOS arrived on the day of the meeting and thanks to Keith Dix having some disc drives in his car, it was got up and running at the meeting. The general opinion of it was very good, and Mark Langdon is now working out how practical it would be for us (actually him!) to produce it for us here in New Zealand, and how much it would cost to do so. The other alternative is just to purchase them for members straight from England. They would cost around \$300.00 for the interface and DOS. The interface has a spare socket especially to fit another 4K, as this is apparently what it needs to work. I haven't found out whether they are selling this 4K eprom/chip in England though, or whether we have to put them in ourselves if we



```

/ REM RHOON - HOLLAND
30 CLEAR200:DIMU(5):M=1500:A$="COLOUR GENIE"
40 S=0:ZZ=0
50 GOSUB940
60 IFM>1500THEN110
70 COLOUR1:PRINT@523,"DO YOU WANT INSTRUCTIONS (Y OR N)?"CHR$(14):
80 GOSUB1180
90 IFI$="Y"THENPRINTCHR$(15)::GOSUB1010:GOSUB940:GOTO110
100 IFI$<>"N"THEN80
110 COLOUR4:PRINT@523," CHOICE OF DIFFICULTY (1-9)? "CHR$(14):
120 GOSUB1180
130 IFASC(I$)<49ORASC(I$)>57THEN120
140 R=VAL(I$):PRINTCHR$(15)
150 E=R/400:D=R/800
160 P=17808+R*40
170 O=P:Q=O+40:Z=O-8
180 CLS:COLOUR4
190 PRINT@931,"I N T R U D E R S"
200 COLOUR12
210 FORI=1TO23:POKE(18328+RND(38)),246:POKE(18328+RND(38)),246:PRINT:FORN=1TO25
:NEXT:NEXT
220 S=S+1:H=1:G=0
230 COLOUR12
240 PLAY(2,1,1,10):PLAY(3,1,2,10)
250 IFPEEK(P)=246THEN300ELSEPOKEP,192:T=P
260 IFPEEK(&HF840)=32THENP=P-1:H=H+EELSEIFPEEK(&HF840)=64THENP=P+1ELSEH=H+D
270 IFP<0THENP=TEELSEIFP=QTHEN430
280 F=INT((P-O)/7)+1:K=P-Z-F*7:PLAY(1,F,K,10)
290 POKET,32:POKE(18328+RND(38)),246:POKE(18328+RND(38)),246:PRINT:GOTO250
300 PLAY(1,1,1,0):PLAY(2,1,1,0):PLAY(3,1,1,0)
310 SOUND0,255:SOUND1,15:SOUND2,255:SOUND3,15:SOUND4,255:SOUND5,15:SOUND6,23:SOU
ND7,7:SOUND8,31:SOUND9,31:SOUND10,31:SOUND11,255:SOUND12,255:SOUND13,9
320 POKEP,243:POKEP-41,227:POKEP-39,223:POKEP+41,227:POKEP+39,223
330 FORB=0TO255:OUT255,B:NEXT
340 POKEP,246:POKEP-39,32:POKEP-41,32:POKEP+41,32:POKEP+39,32
350 LGR
360 COLOUR2:PRINT@21,"SPACESHIP NR."S
370 PRINTTAB(24)"DESTROYED"
380 U(S)=INT(.6*(P-17500)/H)
390 PRINT@0,"POINTS:"U(S)
400 ZZ=ZZ+1
410 FORN=1TO1000:NEXT
420 GOTO480
430 PLAY(2,1,1,0):PLAY(3,1,1,0):FORN=1TO12:PLAY(1,6,N,15):FORN=1TO10:NEXT:NEXT:
PLAY(1,1,1,0)
440 COLOUR2:PRINT@21,"SPACESHIP NR."S
450 PRINTTAB(21)"HAS MADE IT"
460 U(S)=INT(.8*(P-17500)/H)
470 PRINT@0,"POINTS:"U(S)
480 IFS=5THEN530
490 COLOUR8:PRINT@920,"PRESS (RETURN) TO CONTINUE":
500 GOSUB1180
510 IFASC(I$)<>13THEN500
520 POKEP-1,32:PRINT@920,CHR$(30)::P=0:GOTO220
530 COLOUR11:PRINT@920,"PRESS (RETURN) FOR SUMMARY":
540 GOSUB1180
550 IFASC(I$)=13THEN570
560 GOTO540
570 CLS:COLOUR6
580 PRINTTAB(15)"INTRUDERS"
590 COLOUR12:PRINT:PRINT
600 PRINTTAB(9)"SPACESHIP POINTS":PRINT

```

```
610 COLOUR12
620 FORS=1TO5
630 PRINTTAB(13)CHR$(35)S:PRINTTAB(24)U(S)
640 NEXT
650 V=U(1)+U(2)+U(3)+U(4)+U(5)
660 IFZZ=5THENV=0
670 PRINT:PRINTTAB(11)" TOTAL : "STRING$(6,32)V
680 IFZZ=5GOSUB1140
690 L=0:IFV>MTHENM=V:L=1
700 COLOURRND(15):PRINT@566,STRING$(27,127)
710 FORN=606TO766STEP40:PRINT@N,CHR$(127):PRINT@N+26,CHR$(127):NEXT
720 PRINT@806,STRING$(27,127)
730 COLOUR4:PRINT@650,"HIGHEST SCORE:"M
740 IFL=1THENGOSUB860
750 IFL=0THENT=730+((19-LEN(A$))/2)ELSET=730
760 FORN=1TO2:FORNN=1TO7:NEXT
770 COLOURRND(15):PRINT@T,A$
780 COLOUR8:PRINT@929,"ANOTHER GAME (Y/N)? ":IFN=1THENPRINTCHR$(127):ELSEPRINTC
HR$(32):
790 I$=INKEY$:IFI$=""OR(I$<>"N"ANDI$<>"Y")THENNEXT:GOTO760
800 IFI$="Y"THEN40
810 CLS
820 GOSUB940
830 COLOUR8:PRINT@495,"END OF GAME"
840 PRINT@800,:COLOUR1
850 END
860 COLOUR14:PRINT@730,STRING$(19,95)STRING$(19,24):A$=""
870 PRINTCHR$(14):
880 GOSUB1180
890 J=ASC(I$):IFJ=13THENPRINTSTRING$(19-LEN(A$),32)CHR$(15):RETURN
900 IFJ=8ANDLEN(A$)>1THENA$=LEFT$(A$,LEN(A$)-1):PRINTCHR$(95)STRING$(2,24)CHR$(9
5)CHR$(24):GOTO880
910 IFJ=8ANDLEN(A$)=1THENA$="":PRINTCHR$(95)STRING$(2,24)CHR$(95)CHR$(24):GOTO8
80
920 IFJ=32OR(J>64ANDJ<123)THENPRINTI$:A$=A$+I$:IFLEN(A$)>18THENPRINTCHR$(15):R
ETURN
930 GOTO880
940 CLS:COLOURRND(15)
950 PRINT@200,STRING$(41,127):
960 FORN=1TO10:PRINTTAB(39)STRING$(2,127):NEXT
970 PRINTSTRING$(39,127)
980 COLOUR6:PRINT@335,"INTRUDERS"
990 COLOUR15:PRINT@414,STRING$(11,220)
1000 RETURN
1010 CLS:COLOUR6:PRINTTAB(15)"INTRUDERS"
1020 COLOUR11
1030 PRINT:PRINT:PRINT"Far away in the asteroid belt,"
1040 PRINT"a small group of people has started a"
1050 PRINT"mining company."
1060 PRINT:PRINT"You must from time to time supply these"
1070 PRINT"people with food, in your spaceship."
1080 PRINT:PRINT"There are 5 spaceships you have to"
1090 PRINT"bring to the other side in the shortest"
1095 PRINT"possible time"
1100 PRINT:PRINT"The controls are <"CHR$(253)"> and <"CHR$(255)">."
1110 COLOUR8:PRINT@920,"PRESS (RETURN) TO START THE GAME":
1120 GOSUB1180
1130 IFASC(I$)=13THENRETURNELSE1120
1140 COLOUR13:PRINT@566,"SORRY YOU HAVE NOT SURVIVED"
1150 FORN=7TO1STEP-1:FORNN=1TO45:PLAY(1,1,N,15):NEXT:NEXT
1160 PLAY(1,1,1,0)
1170 RETURN
1180 I$=INKEY$:IFI$=""THEN1180
1190 RETURN
```

### CHANGE PRINT TO LPRINT

-----

```

5000 FOR I=22528 to 32767
5001 IF PEEK(I)=178 THEN POKE I,175
5002 NEXT I

```

To start this program, type RUN 5000

This program was given to us by Dai Bevan, of Wellington.

### NEWSLETTERS

-----

Some of you may have missed out on the last newsletter. If this is the case, please get in touch with us, and we will send it out with your next one.

### BITS & PIECES FROM BRYCE NICHOL

-----

Here are some oddities in games which might prove useful.

1. To pause the action in BAK PAK, hit the up arrow, '1' and '2' keys at the same time. You can restart with RETURN. This combination of keys breaks into basic programs.

2. In EAGLE press SHIFT or put down LOCK to get ashield. The game will not get any harder, but nothing will be able to hit you.

3. When you get hit in WURM the screen flashes temporarily. If you hit both reset buttons while it is doing this, all of the graphic characters will change to coloured blocks until you get hit again.

At around 32,000 points WURM crashes. The points for the Worm and the mushrooms are subtracted from your score, and the game stays at first to sixth screen difficulty (no ducks or spiders.)

From theseventh screen on, WURM falls into a repeating pattern. There is a grey worm, followed by white, cyan, purple, dark blue, light yellow, greenish yellow, and red worms. After the red worm, comes alternately an invisible or grey worm and then the pattern repeats again. The screen border colour is the same as the colour of the next worm.

When all of your lives are gone and the screen comes up where you press RETURN to get back to the title screen, try putting in your initials or name (up to 4 letters). This will come up alongside your score in the record. But funny things happen if you press CLEAR or any of the function keys instead.



making System tapes out of basic programs.

#### TAPE RECORDERS

-----  
I have seen a new recorder, a small white one, that works very well with the Colour Genie. It is available from either Dick Smith or David Reid Electronics for \$69.95, which is a very reasonable price. It is probably about 2/3rds the size of the EACA recorder, but it has all that the EACA has on it (tape counter, remote doohicky etc) and also has a couple of extra items. I think it is called 'Euroglider' but will find out for sure by the next meeting.

#### NEXT MEETING

-----  
This will be on Monday, the 27th May, 1985. See you all there.

Nola Huggins.

```

1 REM      THIS IS THE SECOND PROGRAM THAT HERMAN PHILIPSEN
           HAS TRANSLATED FOR US FROM A DUTCH PROGRAM
2 REM
3 REM
4 REM
5 REM
6 REM
10 CLS:COLOURS
20 PRINT@455,"MOONLANDING"
30 FORZ=0TO383
40 READZ1
50 POKE&HF400+Z,Z1
60 NEXT
70 DATA0,0,1,3,7,15,15,15,35,36,255,127,63,31,255,255,196,36,255,254,252,248,255
,255,0,0,128,192,224,240,240,240,15,7,7,63,127,255,255,127,255,255,252,252,252,2
53,252,253,255,255,63,63,63,191,63,191,240,224,224,252,254,255,255,254
80 DATA 31,63,63,63,127,63,63,4,252,253,252,253,252,255,255,3,63,191,63,191,63,2
55,255,192,248,252,252,252,254,252,252,32,8,16,32,240,0,0,0,0,3,7,15,0,0,0,0,1
92,224,240,0,0,0,0,0,16,8,4,15,0,0,0,0
90 DATA 0,0,0,0,0,0,1,3,0,0,0,0,35,36,255,127,0,0,0,0,196,36,255,254,0,0,0,0,0,0
,128,192,7,15,15,15,15,7,7,63,63,31,255,255,255,255,252,252,252,248,255,255,255,
255,63,63,224,240,240,240,240,224,224,252
100 DATA 127,255,255,127,31,63,63,63,252,253,252,253,252,253,252,253,63,191,63,1
91,63,191,63,191,254,255,255,254,248,252,252,252,127,63,63,4,8,16,32,240,252,255
,255,3,3,7,15,0,63,255,255,192,192,224,240,0,254,252,252,32,16,8,4,15
110 DATA0,0,3,7,15,63,127,255,0,248,254,255,255,255,255,255,0,0,3,143,255,255,25
5,255,0,0,128,192,224,240,255,255,0,0,0,0,56,127,255,255,0,0,128,192,252,254,255
,255,0,0,0,0,0,1,255,255,0,0,0,0,224,248,254,255
120 DATA2,2,4,4,9,18,18,36,64,32,32,144,144,8,72,68,2,4,4,9,9,16,18,34,64,64,32,
32,144,72,72,36
130 DATA2,136,32,1,64,2,132,1,160,18,0,8,4,64,2,136,145,44,64,1,32,8,2,129,129,1
6,65,2,4,16,64,2

```

```

140:CLS:RANDOM:Z=0:H=8099:V=55:T=2000:F=1000:S=0:M=4500:G=1.62:Z1=1:I=0:Z2=0
150:FORA=1TO20:COLOURRND(15):PRINT@RND(300)+500,".":NEXT
160:COLOUR6:FORA=0TO15
170:PRINT@A,CHR$(211):
180:PRINT@A+360,CHR$(218):
190:NEXT
200:FORA=0TO360STEP40
210:PRINT@A,CHR$(207):
220:PRINT@A+15,CHR$(219):
230:NEXT
240:PRINT@0,CHR$(223):
250:PRINT@360,CHR$(227):
260:PRINT@15,CHR$(227):
270:PRINT@375,CHR$(223):
280:COLOUR11:PRINT@41,"Height:"
290:PRINT@121,"Speed:"
300:PRINT@201,"Engine:"
310:PRINT@281,"Fuel:"
320:COLOUR3
330:PRINT@840,CHR$(165)CHR$(166)CHR$(167):"   ":CHR$(160)CHR$(161)CHR$(162)CHR$(
163)CHR$(164)CHR$(162)CHR$(165)CHR$(166)CHR$(167)CHR$(165)CHR$(164)CHR$(167)CHR$(
160)CHR$(161)CHR$(162)CHR$(163):STRING$(13,"   ):CHR$(160)CHR$(161)CHR$(163)CHR$(
164):
340:PRINTCHR$(162):
350:PRINT@880,STRING$(3,202):CHR$(163)CHR$(164)CHR$(162):STRING$(16,202):CHR$(16
5):STRING$(8,"   ):CHR$(160)CHR$(163)CHR$(164)CHR$(162):STRING$(5,202):
360:PRINT@920,STRING$(23,202):CHR$(161)CHR$(161)CHR$(202)CHR$(161)CHR$(162)CHR$(
202)CHR$(161)CHR$(162):STRING$(9,202):
370:IF$=INKEY$
380:IFI$<>" "THENI=ASC(I$)
390:IFI>47ANDI<58THENS=I-48
400:IFT<=0THENT=0:S=0
410:COLOUR1
420:PRINT@48,USING"#####":H
430:PRINT@129,USING"#####":V
440:PRINT@209,:S
450:PRINT@290,USING"#####":T
460:IFH<100THENZ=0.5ELSEZ=Z1
470:V=V-(F*S/M-G)*Z:H=INT(H-V*Z):T=T-S*Z:M=M-S*Z
480:IFV>50ANDH<1000THENZ=2
490:IFH>4000THENGOSUB1010
500:IFH<4000THENGOSUB1170
510:IFZ2=2THEN740
520:IFS=0THENSOUND6,0:SOUND7,0:SOUND8,0
530:IFS=0ANDH>100THENPRINT@B+161,"   ":
540:IFS>0THENGOSUB1420
550:GOTO370
560:COLOUR1
570:PRINT@B-40,"   ":
580:IFZ=2THENPRINT@B-80,"   ":
590:PRINT@B,CHR$(128)CHR$(129)CHR$(130)CHR$(131):
600:PRINT@B+40,CHR$(132)CHR$(133)CHR$(134)CHR$(135):
610:COLOUR11
620:PRINT@B+80,CHR$(136)CHR$(137)CHR$(138)CHR$(139):
630:PRINT@B+120,CHR$(140)CHR$(141)CHR$(142)CHR$(143):
640:RETURN

```

```

650 COLOUR1
660 IFZ=2THENPRINT@B-40,"      ":PRINT@B-80,"      ":
670 PRINT@B,CHR$(144)CHR$(145)CHR$(146)CHR$(147):
680 PRINT@B+40,CHR$(148)CHR$(149)CHR$(150)CHR$(151):
690 COLOUR11
700 PRINT@B+80,CHR$(152)CHR$(155)CHR$(154)CHR$(155):
710 PRINT@B+120,CHR$(156)CHR$(157)CHR$(158)CHR$(159):
720 IFV<0THENPRINT@B+160,"      ":PRINT@B+163,"      ":
730 RETURN
740 IFV<3THEN770
750 IFV<20THEN810
760 GOTO910
770 COLOUR3:PRINT@859,CHR$(202):CHR$(202):A$=CHR$(193):PRINT@741,A$:A$:PRINT@78
0,A$:A$:A$:COLOUR6:PRINT@740,CHR$(196):COLOUR1:A$=CHR$(219):PRINT@739,A$:PRIN
T@779,A$:PRINT@819,A$:
780 FORC=1TO2:FORA=1TO8:FORB=1TO12:PLAY(1,A,B,11):NEXT:NEXT:NEXT:PLAY(1,1,1,0)
790 PRINT@60,"Perfect landing"
800 GOTO870
810 SOUND0,0:SOUND1,8:SOUND2,0:SOUND3,8:SOUND4,0:SOUND5,8:SOUND6,24:SOUND7,7:SOU
ND8,24:SOUND9,24:SOUND10,0:SOUND11,48:SOUND12,45:SOUND13,3
820 COLOUR1
830 S=3
840 PRINT@60,"Landing too hard !!"
850 PRINT@100,"Landing gear damaged !"
860 GOTO870
870 IFS=2FORA=1TO800:NEXT:PRINT@B,"      ":PRINT@B+40,"      ":PRINT@B+80,"      ":PR
INT@B+120,"      ":CHR$(164):COLOUR11
880 IFS=3THENCOLOUR5:PRINT@B+123,CHR$(227):COLOUR11
890 PRINT@180,"New game (Y/N)":A$=""
900 A$=INKEY$:IFA$="N"THENENDELSEIFA$="Y"THENRESTORE:GOTO140ELSE900
910 SOUND0,255:SOUND1,15:SOUND2,255:SOUND3,15:SOUND4,255:SOUND5,15:SOUND6,23:SOU
ND7,7:SOUND8,31:SOUND9,31:SOUND10,31:SOUND11,255:SOUND12,255:SOUND13,9
920 COLOUR5
930 PRINT@B+120,CHR$(171)CHR$(173)CHR$(174)CHR$(175):COLOUR6
940 PRINT@B,CHR$(172)CHR$(173)" "
950 PRINT@B+40,CHR$(173)CHR$(174)CHR$(175)CHR$(172):COLOUR2:S=2
960 PRINT@B+80,CHR$(173)CHR$(172)CHR$(174)CHR$(175)
970 COLOUR1
980 PRINT@60,"You are now on the"
990 PRINT@100,"moon, that was quick."
1000 FORA=1TO10:CHAR4:FORA1=1TO20:NEXT:CHAR2:NEXT:GOTO870
1010 IFH>7750THENB=65:GOSUB650:RETURN
1020 IFH>7500THENB=65:GOSUB650:RETURN
1030 IFH>7250THENB=105:GOSUB560:RETURN
1040 IFH>7000THENB=105:GOSUB650:RETURN
1050 IFH>6750THENB=145:GOSUB560:RETURN
1060 IFH>6500THENB=145:GOSUB650:RETURN
1070 IFH>6250THENB=185:GOSUB560:RETURN
1080 IFH>6000THENB=185:GOSUB650:RETURN
1090 IFH>5750THENB=225:GOSUB560:RETURN
1100 IFH>5500THENB=225:GOSUB650:RETURN
1110 IFH>5250THENB=265:GOSUB560:RETURN
1120 IFH>5000THENB=265:GOSUB650:RETURN
1130 IFH>4750THENB=305:GOSUB560:RETURN
1140 IFH>4500THENB=305:GOSUB650:RETURN
1150 IFH>4250THENB=345:GOSUB560:RETURN
1160 IFH>4000THENB=345:GOSUB650:RETURN
1170 IFH>3750THENB=385:GOSUB560:RETURN
1180 IFH>3500THENB=385:GOSUB650:RETURN
1190 IFH>3250THENB=425:GOSUB560:RETURN
1200 IFH>3000THENB=425:GOSUB650:RETURN

```

```
1210 IFH>2750THENB=465:GOSUB560:RETURN
1220 IFH>2500THENB=465:GOSUB650:RETURN
1230 IFH>2250THENB=505:GOSUB560:RETURN
1240 IFH>2000THENB=505:GOSUB650:RETURN
1250 IFH>1750THENB=545:GOSUB560:RETURN
1260 IFH>1500THENB=545:GOSUB650:RETURN
1270 IFH>1250THENB=585:GOSUB560:RETURN
1280 IFH>1000THENB=585:GOSUB650:RETURN
1290 IFH>750THENB=625:GOSUB560:RETURN
1300 IFH>500THENB=625:GOSUB650:RETURN
1310 IFH>400THENB=665:GOSUB360:RETURN
1320 IFH>300THENB=665:GOSUB650:RETURN
1330 IFH>200THENB=705:GOSUB560:RETURN
1340 IFH>100THENB=705:GOSUB650:RETURN
1350 IFH>50THENB=745:GOSUB560:RETURN
1360 IFH>40THENB=745:GOSUB650:RETURN
1370 IFH>30THENB=785:GOSUB560:RETURN
1380 IFH>20THENB=785:GOSUB650:RETURN
1390 IFH>0THENB=785:GOSUB560:RETURN
1400 IFH<0THENB=785:GOSUB650:22=2:RETURN
1410 RETURN
1420 COLOUR5
1430 IFPEEK(17569+B)=161THENRETURN
1440 IFV<0THENPRINT@B+201," "
1450 A=RND(4)
1460 IFA=1THENPRINT@B+161,CHR$(168)CHR$(169):
1470 IFA=2THENPRINT@B+161,CHR$(170)CHR$(169):
1480 IFA=3THENPRINT@B+161,CHR$(170)CHR$(171):
1490 IFA=4THENPRINT@B+161,CHR$(168)CHR$(171):
1500 SOUND6,16:SOUND7,55:SOUND8,12
1510 RETURN
```

---



```

0 CLS
1 COLOUR 8:PRINT@450,"KNIGHTSBRIDGE"
2 FOR A=1 TO 1000:NEXT:CLS
3 COLOUR 2:PRINT"ADAPTED FOR THE COLOUR GENIE":PRINT:PRINT"BY TMA
RRETT"
4 PRINT:PRINT"FROM T HARTNALLS BOOK OF COMPUTER GAMES"
5 FOR Z=1 TO 2000:NEXT:CLS
6 COLOUR 6:PRINT"          THE PIECES MOVE THE SAME"
7 PRINT:PRINT"          AS KNIGHTS IN CHESS"
8 FOR A=1 TO 2000:NEXT:CLS
9 COLOUR 1
10 REM KNIGHTSBRIDGE
20 GOSUB 760
30 GOSUB 490:REM PRINT BOARD
40 IF HU=6 OR CO=6 THEN 680:REM
END OF GAME
50 GOSUB 300:REM COMPUTER MOVES
60 GOSUB 490:REM PRINT BOARD
70 IF HU=6OR CO=6 THEN 680:REM
END OF GAME
80 GOSUB 110:REM PLAYER MOVES
90 GOTO 30
100 REM *****
110 REM PLAYER MOVES
120 Q=0
130 M=INT(RND(O)*66)+11
140 Q=Q+1
150 IF Q=500 THEN 680
160 IF H(M)<>72 THEN 130
170 PRINT"YOU MUST MOVE THE PIECE ON"M
180 INPUT N
190 IF N=99 THENQ=500:GOTO 680
200 REM **CHECK IF MOVE LEGAL**
210 P=0
220 CT=1
230 IF M+Z(CT)=N THEN P=1
240 IF CT<8 THEN CT=CT+1:GOTO 230
250 IF P=0 THEN PRINT"ILLEGAL MOVE":
GOTO180
260 IF H(N)=67 THEN HU=HU+1:PRINT"
WELL PLAYED!!!":FOR R=1 TO 500 :NEXT R
270 H(M)=46:H(N)=72
280 RETURN
290 REM *****
300 REM COMPUTER MOVES
310 Q1=0
320 Q1=Q1+1
330 K=INT(RND(O)*66)+11
340 IF Q1=500 THEN 680
350 IF H(K)<>67 THEN 320
360 PRINT"I HAVE TO MOVE THE PIECE ON"K
365 FOR A=1 TO 800:NEXT
370 W=1
380 IF K+Z(W)<11 OR K+Z(W)>77 THEN 400
390 IF H(K+Z(W))=72 THEN PRINT"GOTCHA!!!":CO=CO+1:FOR P=1 TO 1000
:NEXT P:GOTO 450
400 IF W<8 THEN W=W+1:GOTO 380
410 W=1
420 IF (K+Z(W)<11 OR K+Z(W)>77 ) AND W<8THEN W=W+1:GOTO 420
430 IF H(K+Z(W)) <>46 AND W <8 THEN W=W+1:GOTO430
440 IF W=8 AND H(K+Z(W)) <>46 THEN Q1=
500:GOTO 680
450 X=K:Y=K+Z(W)
460 H(X)=46:H(Y)=67

```

```

470 RETURN
480 REM *****
490 REM PRINT BOARD
500 CLS:PRINT:PRINT:PRINT
510 PRINT TAB(8);"My score is"CO
520 PRINT TAB(8);"And yours is"HU
530 PRINT
540 PRINT TAB(8);"1 2 3 4 5 6 7"
550 PRINT TAB(8);"-----"
560 FOR M=70 TO 10 STEP-10
570 PRINT TAB(5);M/10;
580 FOR N=1TO 7
590 PRINT CHR$(H(M+N));" ";
600 NEXT N
610 PRINT M/10
620 NEXT M
630 PRINT TAB(8);"-----"
640 PRINT TAB(8);"1 2 3 4 5 6 7"
650 PRINT
660 RETURN
670 REM *****
680 REM END OF GAME
690 GOSUB 490
700 COLOUR 5:IF HU=6 THEN PRINT"WELL DONE IM BEATEN"
710 COLOUR 4:IF CO=6 THEN PRINT"IM THE WINNER":PRINT"BETTER LUCK
NEXT TIME"
720 IF Q=500 THEN PRINT"I accept your
wish to concede"
730 IF Q1=500 THEN PRINT"I concede to a
Master"
740 END
750 REM *****
760 REM INITIALISE
770 CLS:PRINT"Please stand by...  "
780 DIM H(99),Z(8)
790 X=0:Q1=0:Q=0
800 RANDOM
810 HU=0:CO=0:REM Scores
820 FOR A=1 TO 99
830 IF A>77 OR A=70 OR A=60 OR A=68
OR A=69 OR A=50 OR A=59 OR A=40 OR A=48
OR A=49 THEN 880
840 IF A=30 OR A=38 OR A=39 OR A=20 OR A=28 OR A=29 OR A=18 OR A=
19 OR A<11 THEN 880
850 H(A)=46
860 IF A>70 AND A<78 THEN H(A)=67
870 IF A>10 AND A<18 THEN H(A)=72
880 NEXT A
890 FOR A=1 TO 8:READ Z(A):NEXT A
900 DATA -8,-21,-12,-19,19,12,21,8
910 RETURN

```

## The Colour in the Genie

I recently attacked the nastie of COLOUR 15 which has cropped up some pommie programs. In their systems, COLOUR 15 is white whereas it is black in ours. (They have UHF TV sets you see.) While I was looking into the ROM with Ken Hynds, it occurred to me that a brief discussion of colour may be useful to members.

The colour of a character on the screen is determined by the first four bits of a number when it is converted to binary. In binary, 0 = 0000, 1 = 0001, 2 = 0010, 3 = 0011, and so on up to 15 which is 1111. 16 is 10000 in binary but as the COLOUR only looks at the first four bits of the number, 16 is effectively the same as a 0 to the Genie. The bottom three bits are the colour bits and the top bit is the brightness factor. There are only the three primary colours generated in a TV tube, red, green and blue or RGB for short. A 0 in the COLOUR number turns that colour on and a 1 turns it off. The lowest bit is red, the next is green and the third bit signifies blue. Thus 0110 is bright red (brightness is on if the top bit is a 0), 0101 is bright green and 0011 is bright blue. If two guns are on, we get the complementary colours cyan (0001), magenta (0010) and yellow (0100). 0000 is white because all three guns are on and the brightness is up. 0111 is a dark gray because only the brightness factor is on and 1111 (or 15) is black because everything is off. A 1 at the start of a number shades a colour. If you compare these colours with the COLOUR numbers in the Genie, you will see that they are actually out of order.

All is not lost though as the colour table is stored in RAM at 4390H-439FH and the colour of a particular COLOUR number may be changed by POKEing that location. If you RUN the following program, you will see the colours as they are normally. If you RUN the program again, you will see the colours in the order that I think that they should have been.

```
10 A$ = STRING$(10,127):FOR I = 1 TO 8
20 COLOUR 1:PRINT I;:COLOUR I:PRINT A$;:COLOUR 1:PRINT I+8;:
   COLOUR I+8:PRINT A$:NEXT
30 FOR I = 0 TO 15:READ K:POKE &H4390 + I,K:NEXT
40 DATA 16,5,6,4,7,3,1,2,8,13,14,12,15,11,9,10
```

What is your opinion about this?

To come back to the pommie programs, if it is in BASIC you merely have to POKE &H439F,16 before you RUN it. Unfortunately, machine-code programs do not appear to use the colour table and alterations have to be done with a monitor as before, slowly.

- Allan Clarke

## First Mate - RS232 Driver

Nola has been on to me for some time to write an article describing what the RS232 Driver program does. This is the program you need if you want to communicate with a data-base, bulletin board, mainframe computer, another micro or a serial printer through the serial port on the Colour Genie's right hand side. If the distance between the devices is more than 20-30 metres then you will need a modem as well (about \$200-300). Signals can be sent over the telephone lines with an approved modem. The data that is usually transmitted or received is confined to the printable ASCII characters 32-127. Characters 0-31 are dedicated to computer control functions and characters 128-255 are generally graphics peculiar to a particular machine. This rules out SYSTEM programs in the raw but there are ways of sending and receiving these as ASCII characters. Files that can be sent as is include BASIC programs (no graphics), word processor files (electronic mail) and data-base files. In fact, anything that can be LPRINTed or LLISTed is a probable candidate. Of course, the receiving or transmitting device at the other end has to be set up to the same RS232 protocol as the Genie.

No hardware mod is required for the Genie and it can work at Baud rates from 300 to 4800. We have tried 9600 Baud but the results were erratic. The other Baud rates are reliable. Files can be received to the screen, printer or 16K buffer. Files in the buffer can be viewed, printed, transmitted or converted back to a BASIC program. ASCII files can be prepared for transmission from the buffer with any program that uses LPRINT statements. The driver has been set up to work with the WP2 word processor and also incorporates the printing patch for Super 5/Epson etc printers.

This only touches on the possibilities of this program. One unlikely use I have discovered is that it allows a true MERGE of BASIC programs where lines can be inserted anywhere in another program. By using RENUM as well, one could copy blocks of a program and duplicate them elsewhere in the program. All of this is accomplished at machine code speed.

- Allan Clarke

## WP2 - Word Processor Notes

A user asked me if it was possible to have line lengths of greater than 76 characters, i.e. more than two Genie lines. The short answer is yes. Anyone else who wants longer lines will need to make the following alterations to the program and CSAVE another copy of the program.

Line 36: Add another PRINT statement at the start of the line.

Line 366: Alter A>=G+E to A>G+E+39.

Line 394: Change E=920 and E=880 to E=830 and E=840 respectively.

These edits will allow line lengths of up to 114 characters.

I hope that users have discovered the edit-scroll function with the < and > keys while in the edit mode.

Now for a word about those %'!"#&' tape recorders that won't allow TAPEd text files to be LOAded back into memory. I have not ever had any problem in this respect with the Genies I have used. I do not use a data stabiliser and simply have both MIC and EAR connections into an EACA recorder. I am not saying that there is no problem and I think that it is worth giving a few hints to help those folks who are having difficulty.

DO: Have the playback head aligned properly with a good audio tape.

Keep the head clean using a cotton bud and isopropyl alcohol or methylated spirits; good enough for the BBC and good enough for me.

Try a different tape machine.

Adjust the playback level to 2.5 before starting a LOAD.

Use ordinary ferric tapes (I always use TDK C60 tapes).

Try removing the MIC plug before LOAding and/or removing the EAR plug before TAPEing.

DO NOT: Use head cleaning tapes or demagnetising cassettes.

Adjust the playback level after you've started a LOAD.

Use your computer on the same house wiring going to a refrigerator or washing machine; induction motors are deadly for spikes!

Use C90 tapes. They stretch and the oxide coating is too thin.

I am investigating the 'problem' of data tapes when PRINT#-1 and INPUT#-1 statements are used in a program. However, I need some tapes of files on machines that will not INPUT the data back again. I would be grateful if any members who have this problem could lend me their tape together with a copy of what was placed on the tape.

- Allan Clarke

```

10 CLS:PRINT TAB(20)"Table Mortgage Program":PRINT:PRINT
20 'In Vanilla BASIC by A.G. Clarke - April 1985
30 PRINT:INPUT"Principal loaned $";P:P1=0
40 PRINT:INPUT"Rate percent";R:R=R/100:PRINT
50 PRINT"   Do you want to use monthly repayment figure (M) or no. of years (N)?
";
60 GOSUB 220:IF A$="m" OR A$="M" THEN 80
70 PRINT:INPUT"No. of years of loan";N:RP=P*R*((1+R)^N/((1+R)^N-1):GOTO 90
80 PRINT:INPUT"Monthly repayment $";RP:RP=RP*12
90 CLS:Y=1
100 PRINT "Monthly repayment: $"RP/12
110 PRINT"Year Yearly Princ Int Princ Int"
120 PRINT"   Payment Owing Repaid Paid"
130 I=P*R:I1=I1+I:P1=P1+RP-I
140 PRINT Y;TAB(5);INT(RP);
150 PRINT TAB(13);INT(P);TAB(20);INT(I);TAB(26);INT(P1);TAB(33);INT(I1)
160 P=P-RP+I:IF P<1 THEN END
170 Y=Y+1:IF Y=20 THEN GOSUB 210
180 IF P+P*R>RP THEN 130
190 FOR J=1 TO 12:K=P+P*J*R/12:IF K<J*RP/12+1 THEN R=J*R/12:RP=K:GOTO 130
200 NEXT
210 PRINT"Press <RETURN> to continue:"
220 A$=INKEY$:IF A$="" THEN 220
230 PRINT:RETURN
240 REM To print a copy of the results on a printer, change PRINT at the
250 REM beginning of lines 100-150 to LPRINT.

```

The results are shown on the screen like this:

```

Monthly repayment: $ 327.338
Year Yearly Princ Int Princ Int
      Payment Owing Repaid Paid
  1   3928   20000   3600   328   3600
  2   3928   19671   3540   715   7140
  3   3928   19284   3471   1171  10612
  4   3928   18828   3389   1710  14001
  5   3928   18289   3292   2346  17293
  6   3928   17653   3177   3097  20470
  7   3928   16902   3042   3983  23513
  8   3928   16016   2883   5028  26396
  9   3928   14971   2694   6261  29091
 10   3928   13738   2472   7716  31564
 11   3928   12283   2211   9433  33775
 12   3928   10566   1902  11459  35677
 13   3928   8540    1537  13850  37214
 14   3928   6149    1106  16671  38321
 15   3928   3328    599   19999  38920

```

```

1 REM 3-D CUBE
3 REM 1=LEFT,2=RIGHT,3=DOWN,4=UP,5=CLOSER,6=FURTHER,7=INCREASE IN PERSPECTIVE,8=
DECREASE IN PERSPECTIVE
5:CLS:FCLS:FGR:FCOLOUR4
10 N=16
20 DIMP(50),X(50),Y(50),Z(50),A(50),B(50)
40 D=10:P=0.5
50 SI=SIN(0.09):CO=COS(0.09)
60 FORI=1TON
70 READP(I),X(I),Y(I),Z(I)
80 NEXTI
200 GOSUB300
210 I$=INKEY$:IFI$=""THEN210
230 FCLS
240 ON VAL(I$)GOSUB1000,2000,3000,4000,5000,6000,7000,8000
250 GOTO200
300 FORI=1TON
310 A(I)=X(I)*250/(P*Z(I)+D):B(I)=Y(I)*250/(P*Z(I)+D)
320 NEXTI
330 FORI=1TON
340 IFP(I)=4THENPLOT A(I)+80,B(I)+50:M=A(I)+80:N=B(I)+50
345 IFP(I)=5THENPLOT M,NTDM+(A(I)-A(I-1)),N+(B(I)-B(I-1))
348 IFP(I)=5THENM=M+(A(I)-A(I-1)):N=N+(B(I)-B(I-1))
350 NEXTI
360 RETURN
1000 FORI=1TON
1010 X=X(I)*CO-Z(I)*SI
1020 Z=Z(I)*CO+X(I)*SI
1030 X(I)=X:Z(I)=Z
1040 NEXTI
1050 RETURN
2000 FORI=1TON
2010 X=X(I)*CO+Z(I)*SI
2020 Z=Z(I)*CO-X(I)*SI
2030 X(I)=X:Z(I)=Z
2040 NEXTI
2050 RETURN
3000 FORI=1TON
3010 Y=Y(I)*CO+Z(I)*SI
3020 Z=Z(I)*CO-Y(I)*SI
3030 Y(I)=Y:Z(I)=Z
3040 NEXTI
3050 RETURN
4000 FORI=1TON
4010 Y=Y(I)*CO-Z(I)*SI
4020 Z=Z(I)*CO+Y(I)*SI
4030 Y(I)=Y:Z(I)=Z
4040 NEXTI
4050 RETURN
5000 D=D*0.9:RETURN
6000 D=D/0.9:RETURN
7000 P=P*0.9:RETURN
8000 P=P/0.9:RETURN
9000 DATA4,1,1,1,5,1,1,-1,5,-1,1,-1,5,-1,1,1,5,1,1,1
9010 DATA5,1,-1,1,5,1,-1,-1,5,-1,-1,-1,5,-1,-1,1,5,1,-1,1
9020 DATA4,1,-1,-1,5,1,1,-1,5,-1,-1,-1,5,-1,1,-1
9030 DATA4,-1,1,1,5,-1,-1,1

```